

# 1.

## Operativsystemets brugerfaciliteter

### Indhold:

- 1.1 DOS kommandoer  
(indgår indtil videre ikke i notesamling)
- 1.2 DOS batch programmering
  - a) Note: Dos batch programmering
- 1.3 Systemtilpasning
  - a) Note: Eksempler på config.sys og autoexec.bat

## 1.1.a

## Oversigt over DOS kommandoer

(De med \* markerede kommandoer forventes at kunne anvendes middelbart af de studerende på Datamatiker uddannelsens første semester. De øvrige forventes kendt)

<Ansi.sys> *	<Erase>	<Nlsfunc>
<Append>	<Exit> *	<Numlock>
<Attrib> *	<Expand>	<Path> *
<Batch cmds> *	<Fasthelp>	<Pause> *
<Break>	<Fastopen>	<Power>
<Buffers>	<Fc>	<Power.exe>
<Call> *	<Fcbs>	<Print> *
<Cd> *	<Fdisk>	<Prompt> *
<Chcp>	<Files>	<Qbasic>
<Chdir>	<Find> *	<Ramdrive.sys>
<Chkdsk> *	<For> *	<Rd> *
<Chkstate.sys>	<Format> *	<Rem> *
<Choice>	<Goto> *	<Ren> *
<Cls> *	<Graphics>	<Rename>
<Command> *	<Help>	<Replace>
<Config.sys cmds> *	<Himem.sys>	<Restore>
<Copy> *	<If> *	<Rmdir>
<Country> *	<Include>	<ScanDisk>
<Ctty>	<Install>	<Set> *
<Date> *	<Interlnk>	<Setver>
<Dblspace>	<Intrelnk.exe>	<Setver.exe>
<Dblspace Tips>	<Internat. cmds> *	<Share>
<Dblspace.sys>	<Intersvr>	<Shell> *
<Debug>	<Keyb> *	<Shift> *
<Defrag>	<Label> *	<Sizer.exe>
<Del> *	<Lastdrive>	<Smartdrv>
<Deltree>	<Lh>	<Smartdrv.exe>
<Device>	<Loadfix>	<Sort> *
<Device drivers>	<Loadhigh>	<Stacks>
<Devicehigh>	<Md> *	<Submenu>
<Dir> *	<Mem>	<Subst>
<Diskcomp>	<Memmaker>	<Switches>
<Diskcopy> *	<MenuColor>	<Sys>
<Display.sys> *	<MenuDefault>	<Time> *
<Dos>	<MenuItem>	<Tree> *
<Doskey>	<Mkdir>	<Type> *
<Dosshell>	<Mode Commands>	<Undelete>
<Driver.sys>	<More> *	<Unformat>
<Drivparm>	<Move>	<Ver>
<Echo> *	<Msav>	<Verify>
<Edit>	<Msbackup>	<Vol>
<EGA.SYS>	<Mscdex>	<VSafe>
<Emm386>	<Msd>	<Xcopy> *
<Emm386.exe>	<Multi-config>	

## 1.2.a

# Dos batch programmering

Bjørk Busch

BAT-programmer er alm. tekstfiler (ascii) der bliver fortolket af kommandofortolkeren direkte. De oversættes således ikke først som f.eks. et turbopascal-program og adskiller sig herved fra programmer med extension COM og EXE. De startes dog fra DOS på samme måde, ved at angive navnet på filen (extension ikke nødv.).

BAT-programmer består primært af opstart af alm. programmer, der startes på samme måde som direkte i DOS-SHELLEN.

Fordelen er at man kan lave en sekvens af kommandoer på forhånd, hvilket er interessant ved opgaver der hyppigt gentages.

Der findes udover de alm. DOS-kommandoer nogle ekstra muligheder således at BAT-programmerne kan gøres fleksible. Der er således mulighed for at overføre informationer (argumenter/parameter) til BAT-programmet ligesom der findes enkelte muligheder for at afbryde og styre udførelsen af kommandoer.

**Eksempel på et BAT-program uden parameter (DOKBKUP.BAT):**

```
1: @ECHO OFF
2: REM RUTINE TIL SIKKERHEDSKOPIERING AF DOKUMENTER
3: CLS
4: ECHO INDSÆT DISKETTE TIL DOKUMENT BACKUP I DREV A:
5: ECHO OG TRYK PÅ EN TAST (CTRL+C for afbryd)
6: PAUSE > NUL
7: MD A:\DOKBKUP
8: COPY C:\DOK1\*.WP* A:\DOKBKUP\*. * > A:\DOKBKUP.LOG
9: COPY C:\DOK2\*.WP* A:\DOKBKUP\*. * >> A:\DOKBKUP.LOG
10: ECHO KOPIERING AFSLUTTET >> A:\DOKBKUP.LOG
11: ECHO KOPIERING AFSLUTTET
```

NB Linienumre medtages ikke i programfilen.

BAT-programmet startes ved at skrive:

```
DOKBKUP
```

Forklaring til rutinen:

- 1: normalt vil kommandolinier blive udskrevet før udførelsen. Denne linie bevirker at DE EFTERFØLGENDE ikke vil blive udskrevet FØR udførelsen.  
For at denne kommando ikke selv udskrives sættes der et @ foran kommandoen.
- 2: Der kan (og bør) indlægges kommentar i BAT-programmer, disse indledes med REM og vil ikke blive udført.
- 3: Skærmen blankes
- 4+5: Udskrift af tekst på skærmen. Hvis der ikke var ECHO OFF ville først kommandoen blive udskrevet (ECHO + tekst) og herefter teksten.
- 6: BAT-program standses. Bemærk at den normale tekst på engelsk bliver undertrykt ved at redigere den over i den logiske enhed NUL
- 7: Katalog oprettes på diskette drev A. Hvis det allerede findes vil der fremkomme en meddelelse.
- 8: Filer kopieres og kopieringsmeddelelser omdirigeres til filen A:\DOKBKUP.LOG (hvis den findes vil den blive overskrevet)
- 9: Der kopieres igen filer til disketten og kopieringsmeddelelser omdirigeres til LOG-filen, idet der skrives videre på den eksisterende.
- 10: Afslutningsmeddelelse tilføjes til LOG-filen.
- 11: Afslutningsmeddelelse udskrives på skærmen.

Eksempel på BAT-program med parameter (DELBAK.BAT):

```
1: DEL %1\*.BAK
```

Eksempler på brug:

- a: DELBAK  
DEL \\*.BAK  
Sletter i root på aktuelt drev
- b: DELBAK .  
DEL \\*.BAK  
Sletter i aktuelt katalog på aktuelt drev
- c: DELBAK A:  
DEL A:\\*.BAK  
Sletter i root på A:
- d: DELBAK PASCAL  
DEL PASCAL\\*.BAK  
Sletter i PASCAL underkatalog på aktuelt drev

**NB!** Det skal bemærkes at %0 angiver navn på programfilen, der startes op.

**Eksempel på BAT-program med brug af SHIFT kommandoen (DIR2.BAT):**

```
1: DIR %1
2: DIR %2
3: SHIFT
4: DIR %1
5: DIR %2
```

Eksempel på start:

```
VIS2 *.TXT *.DOK
```

bliver til:

```
1: DIR *.TXT
2: DIR *.DOK
4: DIR *.DOK
5: DIR
```

Når SHIFT anvendes rykkes parametre en plads, herved kan nye nåes men den foreste mistes.

Hvis man ønsker at gemme parametre kan SET kommandoen anvendes.

**Eksempel på brug af IF og GOTO i batfiler (FMT.BAT):**

```
1: @ECHO OFF
2: IF "%1" == "a:" GOTO OK
3: IF "%1" == "A:" GOTO OK
4: IF "%1" == "b:" GOTO OK
5: IF "%1" == "B:" GOTO OK
6: ECHO Der skal angives drevnavn A: eller B:
7: GOTO SLUT
8: :OK
9: C:\DOS\FORMAT.COM %1 %2 %3 %4 %5 %6 %7 %8 %9
10: :SLUT
```

Eksempel på start:

```
FMT A:
```

```
Her vil A: blive formateret.
```

```
FMT eller
```

```
FMT C:
```

```
Her vil der komme en meddelelse (linie 6) og der vil ikke blive formateret.
```

Forklaring:

- 1-5: Her aftestes 1. parameteren. Da denne kan være "tom" (ingen parameter) er det nødvendig at tilføje 1 eller flere tegn så der ikke kommer til at stå IF == hvilket vil give syntaksfejl. Denne udvidelse med ekstrategn medtages så også i det der sammenlignes med.
- 9: FORMAT program startes, idet op til 9 parametre kan overføres (hvis flere ignoreres resten).

**Eksempel på løkke i batfiler (VISFLERE.BAT):**

```
0: @ECHO OFF
1: IF "%1" = "" GOTO VEJL
2: DEL DIRFIL$.$$$
3: :IGEN
4: DIR %1 | FIND "." >> DIRFIL$.$$$
5: SHIFT
6: IF NOT "%1" = "" GOTO IGEN
7: SORT < DIRFIL$.$$$ | MORE
8: DEL DIRFIL$.$$$
9: GOTO SLUT
10: :VEJL
11: ECHO Angiv mindst een parameter Eks: %0 *.COM
12: :SLUT
```

**Eksempel på start:**

```
VISFLERE *.COM *.EXE
Viser COM og EXE filer sorteret.
```

**Eksempel på test om en fil findes (AKOPI.BAT):**

```
1: IF "%1" = "" GOTO SLUT
2: IF EXIST %1 GOTO KOPIER
3: ECHO %1 findes ikke
4: GOTO SLUT
5: :KOPIER
6: COPY %1 A:\
7: :SLUT
```

**Eksempel på start:**

```
AKOPI MINFIL.TXT
```

**Forklaring:**

Det er nødvendigt at 1. parametren ikke er tom i linie 2, da der ellers vil komme syntaksfejl.

**Eksempel på test af returkode fra program:**

Programmer har mulighed for at give en returkode tilbage til DOS, som så kan anvendes i BAT-programmering.

Der er enkelte DOS kommandoer der udnytter denne mulighed men de fleste gør ikke.

```
1: MITPROG
2: IF ERRORLEVEL 4 GOTO LEV4
3: IF ERRORLEVEL 1 GOTO LEV1
4: :LEV0
.: .....
.: :LEV1
.: .....
.: :LEV4
```

Bemærk der også hoppes hvis ERRORLEVEL er større. Det er derfor en fordel at teste i faldende orden.

**Brug af systemvariable:**

Det er muligt at gemme værdier i parameterblokken (environment).

Disse variable kan opsættes direkte fra DOS eller fra BAT-programmer og kan også bruges til at returnere værdier fra et BAT-program.

Der er 2 måder at anbringe værdier i parameterblokken. Den ene er med PATH kommandoen, hvor søgestien for programmer kan opsættes (tilsvarende med APPEND), den anden måde er med SET kommandoen. Ved opstarten indlægges desuden navnet på shellen.

Man kan få en oversigt over systemvariablerne med kommandoen SET uden parametre. Systemvariable kan indeholde små bogstaver og specialtegn.

**Eksempel på brug systemvarabel med SET:**

- a: SET DOSDIR=C:\DOS  
Der oprettes en variabel med navnet DOSDIR og indholdet "C:\DOS". Denne linie kunne f.eks. indlægges i AUTOEXEC.BAT.
- b: SET GEMP1=%1  
Denne linie tænkes indlagt i et BAT-program, der startes med parametre. 1. parametren vil her blive gemt i variabelen GEMP1 som så senere kan anvendes.
- c: SET GEMP1=  
Her slettes variabelen GEMP1 igen. Det er vigtigt at der ikke er blanke efter lighedstegnet, da variable godt kan indeholde blanke. PAS PÅ!!!  
Linien kunne ligge til slut i samme BAT-program som oprettede variabelen (oprydning).
- d: SET GEMPATH=%PATH%  
Her oprettes en variabel med den aktuelle søgesti som indhold.

**Eksempel på BAT-program med brug af systemvariable (DIR2KAT.BAT):**

```
1: SET P1=%1
2: :IGEN
3: SHIFT
4: IF "%1" = "" GOTO SLUT
5: DIR %P1%\%1
6: GOTO IGEN
7: :SLUT
8: SET P1=
```

**Eksempel på start:**

```
DIR2KAT A: PASCAL WP51
Der bliver lavet 2 DIR kommandoer.
1. DIR A:\PASCAL
2. DIR A:\WP51
```

I det efterfølgende vises andre eksempler på BAT-programmer.

#### WP.BAT

```
1:  REM Opstart til wordperfect med diverse parameter.
2:  PUSHDIR                      shareware prog.
3:  E:
4:  CD E:\WP42
5:  WP /D-D:\WPDOK /M-E:\WP42\STARTOP %1 %2 %3 %4 %5 %6 %7 %8 %9
6:  CD E:\
7:  CD D:\
8:  POPDIR                       shareware prog.
```

#### MTXBKUP.BAT

```
1:  REM Sikkerheds kopiering af system.
2:  ECHO OFF
3:  CLS
4:  ECHO Indsæt diskette til BACKUP af mattext-dokumenter i DREV A
5:  ECHO Tast herefter RETURN - Der kan fortrydes med CTRL + C
6:  PAUSE > NUL
7:  MD A:\DOK1
8:  MD A:\DOK2
9:  COPY D:\MATTEXT\DOK1\*.MTX A:\DOK1\*. * /V
10: COPY D:\MATTEXT\DOK2\*.MTX A:\DOK2\*. * /V
11: ECHO BACKUP afsluttet
```

#### DIREXT.BAT

```
1:  REM Filoversigt sorteret på EXTENTION
2:  DIR %1 %2 | SORT /+10 | FIND "." | FIND " 0 " /V | MORE
```

#### CP.BAT

```
1:  @ECHO OFF
2:  REM Opsæt tegntabel for tastatur og skærm
3:  IF NOT '%1'==' ' GOTO SETCODE
4:  ECHO Angiv nr. på tegntabel eks.      %0 865
5:  GOTO SLUT
6:  :SETCODE
7:  KEYB DK,%1,C:\BIN\KEYBOARD.SYS
8:  MODE CON CODEPAGE PREPARE=((%1)C:\BIN\EGA.CPI)
9:  MODE CON CODEPAGE SELECT=%1
10: :SLUT
```



SETSTI.BAT

```
1:  REM Tilføj kataloger til default path
2:  @ECHO OFF
3:  PATH \;E:\BATFILER;E:\UTILITY;%DOS%
4:  :LOOP
5:  IF '%1'== '' GOTO SLUT
6:    PATH %1;%PATH%
7:    SHIFT
8:  GOTO LOOP
9:  :SLUT
10: PATH
```

ASM.BAT

```
1:  REM Rutine til assemblering og linkning - skrabet udgave
2:  @ECHO OFF
3:  D:\TASM\TASM %1,, /mu
4:  IF ERRORLEVEL 1 GOTO SLUT
5:    D:\TASM\TLINK %1
6:  IF ERRORLEVEL 1 GOTO SLUT
7:    ECHO Program %1 er assembleret og linket fejlfrit
8:  :SLUT
```

ASM2.BAT

```
1:  REM Rutine til assemblering og linkning - udbygget udgave
2:  @ECHO OFF
3:  IF '%1'=='' GOTO VEJL
4:  IF EXIST %1.ASM GOTO TASM
5:  ECHO %1.ASM findes ikke
6:  :VEJL
7:  ECHO Der skal angives filnavn uden .ASM
8:  ECHO Eks.      %0 ASMPROG
9:  GOTO SLUT
10:
11:  :TASM
12:  ECHO Assemblering af %1.ASM startet
13:  D:\TASM\TASM %1,, /mu
14:  IF ERRORLEVEL 1 GOTO ASMERR
15:  ECHO Link af %1.OBJ startet
16:  D:\TASM\TLINK %1
17:  IF ERRORLEVEL 1 GOTO LINKERR
18:  ECHO Program %1 er assembleret og linket fejlfrit
19:  GOTO SLUT
20:
21:  :ASMERR
22:  ECHO Fejl i assemblering af %1
23:  GOTO SLUT
24:  :LINKERR
25:  ECHO Fejl i linkning af %1
26:
27:  :SLUT
```

123.BAT

```
1: REM Opstart af lotus-123 i eget katalog uafhængig af drev
2: IF EXIST E:\123\123start.BAT GOTO EDREV
3: IF EXIST D:\123\123start.BAT GOTO DDREV
4: GOTO EJINST
5: :EDREV
6: E:
7: \123\123start.BAT %1 %2 %3 %4 %5 %6 %7 %8 %9
8: GOTO SLUT
9: :DDREV
10: D:
11: \123\123start.BAT %1 %2 %3 %4 %5 %6 %7 %8 %9
12: GOTO SLUT
13: :EJINST
14: ECHO Lotus 123 er ikke installeret
15: :SLUT
```

123START.BAT

```
1: @ECHO OFF
2: IF "%1"=="?" GOTO VEJL
3:
4: IF "%S123%"==" " SET S123=VGA          Sæt default første gang
5: IF NOT "%1"==" " SET S123=?          Hvis parameter
6: IF "%1"=="V" SET S123=VGA
7: IF "%1"=="V25" SET S123=VGA13225
8: IF "%1"=="V28" SET S123=VGA13228
9: IF "%1"=="V44" SET S123=VGA13244
10: IF "%1"=="V60" SET S123=VGA8060
11:
12: IF NOT %S123%=="?" GOTO START
13:
14: :VEJL
15: ECHO -----
16: ECHO Parametre for opstart af Lotus 123
17: ECHO V VGA 80 x 25
18: ECHO V25 VGA 132 x 25
19: ECHO V28 VGA 132 x 28
20: ECHO V44 VGA 132 x 44
21: ECHO V60 VGA 80 x 60
22: ECHO -----
23: GOTO SLUT
24:
25: :START
26: PUSHDIR
27: CD \123
28: 123 %S123%          aktuelt drev forudsættes
29: CD \                start med udvalgt driver %S123%
30: POPDIR
31:
32: :SLUT
```

SPILINST.BAT

```
1: @ECHO OFF
2: IF '%1'==' ' GOTO VEJL
3: IF '%2'==' ' GOTO VEJL
4: IF EXIST %2.ARJ GOTO INSTARJ
5: IF EXIST %2.ZIP GOTO INSTZIP
6: IF EXIST %2.LZH GOTO INSTLZH
7:
8: ECHO Fejl: filen %2 findes ikke
9: ECHO -----
10: FOR %%F IN (*.ARJ *.ZIP *.LZH) DO ECHO %%F
11: GOTO SLUT
12:
13: :VEJL
14: ECHO Instalation af pakket spil.
15: ECHO Der skal startes fra drev / katalog med pakket spil.
16: ECHO Der skal være path til udpakke-programmet:
17: ECHO ARJ.EXE / PKUNZIP.EXE / LHARC.EXE
18: ECHO -----
19: ECHO Instalationsrutinen opstartes med følgende parametre.
20: ECHO 1. d1:\path (drev og katalog for spil)
21: ECHO 2. spil (navn spilfil uden .ARJ / .ZIP / .LZH)
22: ECHO eksempel
23: ECHO 1. %0 C:\SPIL KEEN
24: ECHO 2. %0 D:\ KEEN
25: ECHO -----
26: ECHO Tryk på en tast for oversigt over pakkede spil
27: PAUSE > NUL
28: FOR %%F IN (*.ARJ *.ZIP *.LZH) DO ECHO %%F
29: GOTO SLUT
30:
31: :INSTARJ
32: @ECHO ON
33: MD %1\%2
34: ARJ x -r %2.ARJ %1\
35: REM ARJ x -r %2.ARJ %1\%2
36: GOTO SLUT
37:
38: :INSTZIP
39: @ECHO ON
40: MD %1\%2
41: PKUNZIP -o %2.ZIP %1\%2
42: GOTO SLUT
43:
44: :INSTLZH
45: @ECHO ON
46: MD %1\%2
47: LHARC x -p %2.LZH %1\%2
48: GOTO SLUT
49:
50: :SLUT
```

ikke nødvendig.

## 1.3.a

# Eksempel på config.sys og autoexec.bat

Bjørk Busch

### CONFIG.SYS

```
1:  DEVICE=C:\XSYS\FASTBIOS.SYS
2:  FILES=30
3:  FCBS=1
4:  BUFFERS=10
5:  LASTDRIVE=E
6:  DEVICE=C:\SYS\HIMEM.SYS
7:  DEVICE=C:\SYS\EMM386.EXE noems
8:  DOS=HIGH,UMB
9:  DEVICEHIGH=C:\XSYS\EANSI.SYS
10: DEVICEHIGH=C:\XSYS\GMOUSE.SYS *4
11: DEVICEHIGH=C:\SYS\DISPLAY.SYS CON:=(EGA,865,1)
12: DEVICEHIGH=C:\SYS\SMARTDRV.SYS 1024
13: COUNTRY=045,865,C:\SYS\COUNTRY.SYS
14: SHELL=C:\SYS\COMMAND.COM C:\SYS /E:512 /P
```

### AUTOEXEC.BAT

```
1:  ECHO OFF
2:  SET DOS=E:\MSDOS
3:  PATH %DOS%
4:  SET COMSPEC=%DOS%\COMMAND.COM
5:  PROMPT $e[41;30m$P$G$e[43;30m      prompt i anden farve
6:  LOADHIGH KEYB DK,865,C:\SYS\KEYBOARD.SYS
7:  MODE CON CODEPAGE PREPARE=((865)C:\SYS\EGA.CPI)
8:  MODE CON CODEPAGE SELECT=865
9:  LOADHIGH FASTOPEN C:=10 D:=50 E:=50
10: CLS
11: DATE
12: TIME
13: D:
14: CALL E:\BATFILER\SETSTI.BAT      se tidligere
15: ECHO Tast CTRL+C hvis virustest ikke skal køre
16: PAUSE > NUL
17: SCAN2 C:
18: PAUSE > NUL
19: SCAN2 E:
```